Lab S3: Linear and Nonlinear Systems

This Maple lab is based in part on earlier versions prepared by Professors R. Falk and R. Bumby of the Rutgers Mathematics department.

Introduction. In this lab we use Maple to find *eigenvalues* and *eigenvectors* of matrices, and solve linear systems of algebraic equations and systems of first order linear differential equations.

Then we will use Maple to study the phase plane of an autonomous nonlinear system of two differential equations, i.e., a system of the form

$$x'(t) = F(x,y), \qquad y'(t) = G(x,y). \tag{1}$$

The distinguishing feature of an autonomous system is that the expressions defining the functions F and G do not contain the independent variable t (our linear systems with constant coefficients are special examples of these). This allows many properties of the solutions to be studied using the curves, called *trajectories*, that show the path in the xy plane followed by the solutions. (It is an easy exercise to show that if an initial condition is on a trajectory, then the whole solution follows that trajectory). The Maple command `DEplot` may be used to draw trajectories and direction fields for such systems.

Please obtain the seed file from the web page and save it in your directory on eden, then prepare the Maple lab according to the instructions and hints in the introduction to Lab 0. Turn in only the printout of your Maple worksheet, after removing any extraneous material and any errors you have made. There is a *supplementary worksheet* for Lab S3 that contains some commands which you are asked to try out but which should not appear in the final lab.

0. Setup. As usual, the seed file begins with commands which load the required Maple packages: `with(plots):` and `with(DEtools):`. In this lab we also require a Linear Algebra package. There are now two such packages in Maple, not completely compatible. The newer package, called `LinearAlgebra` is used here since it has more respect for its user.

1. Matrix entry. The seed file and the supplementary worksheet include the definitions of three matrices, A and B:

```
A:= <<-1|2|3>,<-3|-2|-1>>;
B:= <<1,0,-1>|<3,1,-2>>;
```

The syntax for entering matrices into Maple should be clear from these examples (and the resulting matrices); if not, you can read about it by entering `?<` at the prompt. For practice, enter in your worksheet (i) a column vector with three entries, all different; (ii) a row vector with four entries, all different; and (iii) a 3×3 matrix whose entries are all distinct.

2. Matrix Operations. The goal of this section is to understand how the matrix operations of addition (+), matrix multiplication (.), scalar multiplication ($*$), and powers (ˆ) act in various circumstances. We will experiment with these, and since some of our experiments may give errors and unexpected results, we will do so in the *supplementary worksheet.* The worksheet that you submit should contain only a *discussion*, guided by the questions below.

A few examples of the use of matrix operations are already in the supplementary worksheet, and you should add others to allow a full discussion of these operations. Some of these examples will lead to errors, and such errors will find their place in the worksheet discussion: your comment should include a description of the failed command and your interpretation of the error message.

If you have any doubt about your interpretation of a result, you can consult *Maple help.* There are various ways to do so (see the *Help* button at the top of the worksheet) but here are two particular ones that may be helpful: at a command prompt `>`, typing "?anything" (`> ?anything`) will give help on the topic "anything" (try `> ?LinearAlgebra` or `> ?.`), and placing the cursor on a command you have typed, and pressing F2, will give help on that command.

Here are the questions to guide your discussion:

(1) How is `M1+M2` computed when `M1` and `M2` are matrices? Is it *always defined*? If not, how does Maple indicate that it is not defined?

(2) How is `M1.M2` computed when `M1` and `M2` are matrices? Is it *always defined*? If not, how does Maple indicate that it is not defined?

(3) How is `M^c` computed when `M` is a matrix and `c` is an integer constant (possibly negative)? Is it *always defined*? If not, under what circumstances is it undefined, and how does Maple indicate the nature of the problem?

(4) How are `c*M` and `M*c` computed when `M` is a matrix and `c` is a scalar? What problems can arise when one tries to carry out scalar multiplication using `c.M` or `M.c`? Are these products always defined? Do they always give matrices as answers? Does it make any difference whether the scalar comes first or second in the product? Whether the scalar is a number (constant) or a variable?

3. Eigenvalues, eigenvectors, and matrix exponentials.

3a. Eigenvalues and eigenvectors. The following lines appear in the seed file. They use the `Eigenvectors` command to obtain eigenvectors *and* eigenvalues of three related matrices.

```
M1:=A.B;
M2:=B.A;
M3:=M1 ^ (-1);
(Vals1,Vecs1):=Eigenvectors(M1);
(Vals2,Vecs2):=Eigenvectors(M2);
(Vals3,Vecs3):=Eigenvectors(M3);
```

(1) The matrices `M1` and `M2` are different. Why? What algebraic rule does not hold for matrix multiplication? What is the relation between their eigenvalues? *Note*: The relation between the eigenvalues of `M1` and `M2` is an example of a general theorem, which we do not give.

(2) What is the relation between `M1` and `M3`? What is the relation between their eigenvectors? Between their eigenvalues?

3b. Matrix Exponentials. We will use the matrix exponential function e^{At} to find explicit solutions of differential equations presented in the form $\mathbf{y}' = A\mathbf{y}$ (see Boyce and diPrima Section 7.7; as noted there, e^{At} is the same as the special *fundamental matrix* $\Phi(t)$). The matrix exponential e^{At} satisfies the key equations $d/dt(e^{At}) = Ae^{At}$ and $e^{At}\big|_{t=0} = I$. The `LinearAlgebra` package provides this function for us, and a sample command is included in the work sheet.

To differentiate the matrix exponential, or any matrix or vector function of t, we must use the maple command `map`. (There is also a `Map` function, and the two are not interchangeable. We use only `map` in this lab.) If `M` is a Maple Matrix (or Vector) whose entries depend on the variable t, then the command `map(diff,M,t)` constructs a new Matrix (or Vector) whose entries are obtained by differentiating the entries of `M`.

The following Maple commands, included in the worksheet, construct the fundamental matrix and check that it has all the required properties:

```
Y1:=MatrixExponential(M1,t);
DY1:=map(diff,Y1,t); #derivative of the matrix exponential
MY1:=M1.Y1;
DY1-MY1;# checks equation if zero matrix
subs(t=0,Y2);# checks if initial value is identity matrix
```

To illustrate the connection between the matrix exponential and solutions of individual initial value problems, you should:

(1) use the matrix `Y1` to find the solution of

$$\frac{d\mathbf{y}}{dt} = M_1\mathbf{y} \quad \text{with} \quad \mathbf{y}(0) = \begin{pmatrix} -3 \\ 2 \end{pmatrix}$$

where `M1 := A.B` is the matrix found above.

(2) check that this vector satisfies the differential equations; and check that this vector satisfies the initial conditions.

As discussed in Boyce and DiPrima Section 7.7 (but only partially), when A is an $n \times n$ matrix with n distinct eigenvalues $\lambda_1, \ldots, \lambda_n$ and corresponding eigenvectors $\xi^{(1)}, \ldots, \xi^{(n)}$, the matrix e^{At} can be obtained by the formula $e^{At} = TQ(t)T^{-1}$; here $Q(t)$ is a diagonal matrix whose i^{th} diagonal entry is $e^{\lambda_i t}$ and T is a matrix whose columns are the eigenvectors: $T = (\xi^{(1)} \ldots \xi^{(n)})$. The worksheet includes the instructions

```
Q1 := <<exp(Vals1[1]*t)|0>,<0|exp(Vals1[2]*t)>>;
Y1a := Vecs1.Q1.Vecs1^ (-1); # alternative formula for matrix exponential
Y1-Y1a; # checks alternate formula if zero
```

which construct the matrix in this way (when $A = M_1$), then check that the constructed matrix agrees with the one Maple found with its built-in Matrix Exponential function. You should try out these instructions and understand what they are doing, but no questions are asked about them.

4. Solving systems and identifying equilibria. Consider the matrices

$$M_4 = \begin{pmatrix} 1 & -2 \\ -2 & 1 \end{pmatrix} \quad \text{and} \quad M_5 = \begin{pmatrix} 4 & 0 \\ 3 & 5 \end{pmatrix}. \quad \text{and} \quad M_6 = \begin{pmatrix} -2 & -1 \\ 1 & -1 \end{pmatrix}$$

(1) For each matrix M, use the matrix exponential to solve the equation $d\mathbf{y}/dt = M\mathbf{y}$ with initial conditions

$$(a) \quad \mathbf{y}(0) = \begin{pmatrix} 0 \\ -1 \end{pmatrix}, \quad (b) \quad \mathbf{y}(0) = \begin{pmatrix} 1 \\ 0 \end{pmatrix}, \quad (c) \quad \mathbf{y}(0) = \begin{pmatrix} 2 \\ -1/2 \end{pmatrix}.$$

(2) For each matrix M, by finding its *eigenvalues* determine the type of equilibrium point at the origin: saddle point, stable or unstable node or spiral. Explain briefly.

5. A nonlinear system. We will study the nonlinear system

$$\begin{aligned} x' &= -(x-y)(1-x-y) \\ y' &= x(2+y). \end{aligned} \tag{2}$$

The equilibrium solutions are $[x=0, y=0]$, $[x=-2, y=-2]$, $[x=3, y-2]$, and $[x=0, y=1]$. Maple can obtain these by using the instructions

```
F:=-(x-y)*(1-x-y);
G:=x*(2+y);
eqpts:=solve(F,G,x,y);
```

which are included in the seed file. Also included are instructions to define the differential equations:

```
dex:=diff(x(t),t)=eval(F,{x=x(t),y=y(t)});
dey:=diff(y(t),t)=eval(G,{x=x(t),y=y(t)});
```

For later convenience we have defined F and G to depend on the variables x and y, but in the differential equations we must write these variables as $x(t)$ and $y(t)$; the `eval` command makes this substitution.

b. Trajectories. To study the behavior of the system and the nature and stability of equilibrium points, it is also useful to construct a plot of the phase plane which shows several trajectories as well as the direction field. For the linear systems above we could have used the *exact* solutions that we found to plot some trajectories in the phase plane. However since in general it is not be possible to solve arbitrary systems of nonlinear equations, we use here the *numerical methods* that are part of the `DEplot` command. To produce each trajectory one specifies an initial condition—i.e., a point in the phase plane—and asks Maple to produce the trajectory through that point; to get a significant part of the full trajectory one solves over a range of the independent variable, t, which includes both positive and negative values.

To obtain a good choices of initial conditions, leading to a set of trajectories which illustrate the important features of the phase plane, requires some trial and error. The seed file contains a list of six initial conditions, together with a command to produce a plot of the phase plane containing the direction field and the corresponding six trajectories:

```
inits:=[[x(0)=1,y(0)=1],[x(0)=2,y(0)=1],[x(0)=-4,y(0)=-4],
     [x(0)=3,y(0)=-1],[x(0)=-1,y(0)=-1],[x(0)=1,y(0)=-1]]:
DEplot([dex,dey],[x(t),y(t)],t=trange,inits,window,color=GREEN,
     linecolor=[RED,BLUE,BROWN,PLUM,CORAL,BLACK],thickness=2,stepsize=0.005,
     title="Trajectories:  Nonlinear equation");
```

The choice of stepsize here seems to work well, but you can experiment with other possible choices. Include the resulting plot in your final worksheet.

Discussion: After examining the phase plane plot, try to decide the *type* of each of the four critical points: stable or unstable node, saddle point, or stable or unstable spiral. Give your conclusions for each point in the *discussion* section.

Discussion: In the second *discussion* section you should describe any interesting properties of each of the six trajectories shown. In particular, *for each one*, discuss whether or not that trajectory appears to approach a critical point—and if so, which one—as $t \to \infty$ and as $t \to -\infty$. In your discussion you should identify the various trajectories by color.

c. Linearization. The type and stability of the critical points can be determined by examining the eigenvalues of the corresponding linear system. The matrix of the linearization of the system (1) at the critical point (x_c, y_c) is

$$A := \begin{pmatrix} F_x(x_c, y_c) & F_y(x_c, y_c) \\ G_x(x_c, y_c) & G_y(x_c, y_c) \end{pmatrix}.$$

Maple can find this matrix in a two step process, first constructing the matrix of partial derivatives (which must be done only once), then evaluating it at the critical point (done separately for each point). Instructions for finding A for general x and y and then finding A_1, its value at the critical point at the origin, and the eigenvalues of A_1, are in the seed file:

```
A := Matrix([[diff(F,x),diff(F,y)],[diff(G,x),diff(G,y)]]);
A1 := eval(A,{x=0,y=0});
Eigenvalues(A1);
```

Use Maple to find the matrices A_2, A_3, and A_4, and their eigenvalues, corresponding to the critical points $(-2-2)$, $(3,-2)$, and $(0,1)$ respectively.

Discussion: In a *discussion section*, for each critical point, describe what the eigenvalues you found imply about the type of that point, and compare the answer with the one you found in part **b**. (The answers should agree!)

End of Lab S3